


Satzmaschine

Wenn du im Seriellen Monitor den Buchstaben ‚s‘ eingibst, wird ein zufällig erzeugter Satz angezeigt.

So sieht es aus (Beispiel):



```

Ausgabe Serieller Monitor x
Nachricht (STRG + Enter, um Nachricht an 'Arduino Uno' auf '/dev/ttyA'
Zeilenumbruch 115200 baud

Das Motorrad saust im Zeitlupentempo zum Bahnhof.
Der Bus rollt langsam zum Bahnhof.
Die U-Bahn fährt langsam ans Ende der Welt.
Das Motorrad saust langsam um die Ecke.
Die U-Bahn schleicht langsam zur Haltestelle.
Die U-Bahn saust rasant um die Ecke.
Das Fahrrad schleicht schnell auf die Autobahn.
Der Bus rollt schnell um die Ecke.
Die U-Bahn saust im Zeitlupentempo zum Bahnhof.
Die U-Bahn rast zügig ans Ende der Welt.
Das Motorrad rast langsam zum Bahnhof.
Der Bus fährt zügig um die Ecke.
Das Motorrad saust im Zeitlupentempo um die Ecke.
  
```

Bisher kennst du Arrays, die eine bestimmte Anzahl von Elemente in einer Reihe haben (eindimensionale Arrays).

```
String Eigenschaft[3] = {"schnell ", "langsam ", "im Schrittempo "};
int LED[5] = {3, 4, 5, 6, 7};
```

Außerdem gibt es noch mehrdimensionale Arrays, sie sind in Reihen und Spalten organisiert.

Beispiel für ein zweidimensionales Array:

```
// 4 Reihen, 4 Spalten
int Zahlen[4] [4] =
{
  {1, 2, 3, 4},
  {11, 12, 13, 14},
  {21, 22, 23, 24},
  {31, 32, 33, 34}
};
```

	Spalte 0	Spalte 1	Spalte 2	Spalte 3
Reihe 0	Zahlen[0] [0]	Zahlen[0] [1]	Zahlen[0] [2]	Zahlen[0] [3]
Reihe 1	Zahlen[1] [0]	Zahlen[1] [1]	Zahlen[1] [2]	Zahlen[1] [3]
Reihe 2	Zahlen[2] [0]	Zahlen[2] [1]	Zahlen[2] [2]	Zahlen[2] [3]
Reihe 3	Zahlen[3] [0]	Zahlen[3] [1]	Zahlen[3] [2]	Zahlen[3] [3]

Das Beispielprogramm zeigt das Array an:


```
// 4 Reihen, 4 Spalten
int Zahlen[4] [4] =
{
  {1, 2, 3, 4},
  {11, 12, 13, 14},
  {21, 22, 23, 24},
  {31, 32, 33, 34}
};

void setup()
{
  Serial.begin(115200);

  /*
   Elemente des Arrays anzeigen
   i -> Zeilen, j -> Spalten
   for-Schleife durch die Reihen
  */
  for ( int i = 0; i < 4; i++ )
  {
    Serial.print("Zeile " + String(i) + ": ");

    // for-Schleife durch die Spalten
    for ( int j = 0; j < 4; j++ )
    {
      Serial.print(String(Zahlen[i] [j]) + " ");
    }
    Serial.println();
  }
}
```

```
void loop()
{
  // bleibt leer, das Programm läuft nur einmal
}
```



```
Ausgabe  Serieller Monitor x
Nachricht (STRG + Enter, um Nachricht an 'Arduino Uno' auf '/dev/ttyA
Zeilenumbruch  115200 baud
Zeile 0: 1 2 3 4
Zeile 1: 11 12 13 14
Zeile 2: 21 22 23 24
Zeile 3: 31 32 33 34
```


Weiteres Beispiel:

Das Programm bildet aus dem Array vier Sätze und zeigt sie an.

```
// 2 Reihen, 4 Spalten
/*
  Umlaute im Seriellen Monitor:
  \u00e4 = ä
  \u00f6 = ö
  \u00fc = ü
  \u00df = ß
*/
String Wortliste[2][4] =
{
  {"Der Bus ", "Das Auto ", "Die U-Bahn ", "Das Motorrad "},
  {"h\u00e4lt an der Haltestelle.", "f\u00e4hrt in die Garage.", "stoppt im
  Bahnhof.",
  "flitzt \u00fcber die Autobahn."}
};

void setup()
{
  Serial.begin(115200);
  // alle Sätze anzeigen
  for (int i = 0; i < 4; i++)
  {
    Serial.print(Wortliste[0][i]);
    Serial.println(Wortliste[1][i]);
  }
}

void loop()
{
  // bleibt leer, das Programm läuft nur einmal
}
```



```

Ausgabe Serieller Monitor
Nachricht (STRG + Enter, um Nachricht an 'Arduino Uno' auf '/dev/ttyA
Zeilenumbruch 115200 baud

Der Bus hält an der Haltestelle.
Das Auto fährt in die Garage.
Die U-Bahn stoppt im Bahnhof.
Das Motorrad flitzt über die Autobahn.
  
```

Für diese Anleitung wird nur ein Arduino benötigt.

Definiere die Arrays:

```

// Array für die Artikel vor dem Substantiv und das Substantiv
// jeweils 6 Elemente -> Zählung beginnt mit 0
String Artikel_Substantiv [6] [6] =
{
  {"Der ", "Das ", "Die ", "Das ", "Das ", "Der "},
  {"Bus ", "Auto ", "U-Bahn ", "Fahrrad ", "Motorrad ", "Lastwagen "},
};
// eindimensionales Array für die Eigenschaften
String Eigenschaft [6] = {"rasant ", "im Zeitlupentempo ", "schnell ",
"langsam ", "z\u00fcgig ", "im Schritttempo "};

// Array für Verb und das Objekt
String Verb_Objekt [6] [6] =
{
  {"rollt ", "saust ", "rast ", "schleicht ", "f\u00e4hrt ", "bummelt "},
  {"auf die Autobahn. ", "ans Ende der Welt.", "um die Ecke.", "zum Bahnhof.",
  "zur Haltestelle.", "in die Garage."}
};
// Bereich der zufällig erzeugten Zahlen
int Minimum = 0;
int Maximum = 6;
  
```

Der setup-Teil startet den Seriellen Monitor.

```

void setup()
{
  Serial.begin(115200);

  // auf serielle Verbindung warten
  while (!Serial) {}
}
  
```

Der loop-Teil. Beachte die Kommentare.

```
void loop()
{
  // auf serielle Eingabe warten
  while (Serial.available() > 0)
  {
    // Eingabe im Seriellen Monitor lesen
    char Zeichen = Serial.read();

    // wenn s eingegeben wurde
    if (Zeichen == 's')
    {
      int Position = random(Minimum, Maximum);
      /*
       * Bestimmung des Artikels und des Substantivs
       * entsprechend der Position im Array Artikel_Substantiv
       * 0 -> Artikel
       * 1 -> Substantiv
       */
      Serial.print(Artikel_Substantiv[0] [Position]);
      Serial.print(Artikel_Substantiv[1] [Position]);

      // Bestimmung des Verbs -> Position 0 im Array Verb_Objekt
      Position = random(Minimum, Maximum);
      Serial.print(Verb_Objekt[0] [Position]);

      // Festlegung der Eigenschaft
      Position = random(Minimum, Maximum);
      Serial.print(Eigenschaft[Position]);

      // Objekt auswählen -> Position 1 im Array Verb_Objekt
      Position = random(Minimum, Maximum);
      Serial.println(Verb_Objekt[1] [Position]);
    }
  }
}
```

Hartmut Waller Letzte Änderung: 09.10.22